
4.1. Eng kichik kvadratlarni hisoblash 

Biz II-bobda 𝑌̂  hisoblangan masalasiga to'xtalgandik. Endi biz prognozning aniq 

xatoligini yoki bashorat qilishning aniqligini o'lchash sifatida prognozning standart xatosini 

topamiz. Ishchi katalogingizga o'ting, bo'lim uchun ishchi faylni oching va oziq-ovqat 

xarajatlari ma'lumotlarini joylashtiring. 

version 11.1 

capture log close 

set more off 

log using chap04_food, replace text 

use food, clear 

Keling, haftalik daromadi 2000 dollar bo'lgan uy xo'jaliklari uchun uy -ro'zg'or 

xarajatlarining taxminiy qiymatini topaylik. 

 

4.1.1. Ma'lumotlarni tahrirlash 

41-kuzatuvda income=20 ni kiritib, ma'lumotlarni tahrir qilamiz. Bosqichlar ushbu 

qo'llanmaning 2.5-bo'limida to`liq tushuntirilgan. 

edit 

set obs 41 

replace income = 20 in 41 

 



4.1.2. Regressiyani baholash va keyingi baholash natijalarini olish 

Endi biz quyidagi buyruq orqali oziq-ovqat xarajatlarining regressiyasini 

hisoblaymiz. 

quietly regress food_exp income 

Hisoblashdan keyingi predict buyrug'idan foydalanib, hisoblangan qiymatlarni 

(yhat), eng kichik kvadratlar qoldiqlarini (ehat) topamiz. Bularni hisoblashni biz 2.4.1 -

bo'limda to`liq to`xtalib o`tgandik. 

predict yhat 

predict ehat, residuals 

Biz qo'shadigan yangi buyruq - stdf, bu prognozning standart xatosini hisoblab 

chiqadi. Biz prognoz qilmoqchi bo'lgan yangi kuzatuv topamiz: 

𝑦0 = 𝛽1 + 𝛽2𝑥0 + 𝑒0 

bu yerda 𝑒0  - tasodifiy xato. 𝐸(𝑦0) = 𝛽1 + 𝛽2𝑥0  va 𝐸(𝑒0) = 0 deb faraz qilamiz. 

Shuningdek, biz 𝑒0  ning regressiya xatoliklari bilan bir xil dispersiyaga ega deb faraz 

qilamiz, 𝑣𝑎𝑟(𝑒0) = 𝜎2 va 𝑒0 namuna ma’lumotlarining bir qismi bo‘lgan tasodifiy xatolar 

bilan bog‘liq emas, shuning uchun 𝑐𝑜𝑣(𝑒0, 𝑒𝑖) = 0,  𝑖 = 1,2,… . , 𝑁 . 𝑦0  ning eng kichik 

kvadratlar orqali hisoblanib mos keladigan regressiya chizig'i topiladi: 

𝑦̂0 = 𝑏1 + 𝑏2𝑥0 

Ya'ni, hisoblangan qiymat 𝑦̂0  eng kichik kvadratlar orqali hisoblangan chiziqning 

𝑥 = 𝑥0 nuqtasida berilgan. Ushbu bashoratchi qanchalik yaxshi ishlashini baholash uchun 

biz eng kichik kvadratlar qoldig'iga o'xshash prognoz xatosini aniqlaymiz, 

𝑓 = 𝑦0 − 𝑦̂0 = (𝛽1 + 𝛽2𝑥0 + 𝑒0) − (𝑏1 + 𝑏2𝑥0) 
Ushbu prognoz xatosining taxminiy farqi 

𝑣𝑎𝑟(𝑓) =̂ 𝜎̂2 [1 +
1

𝑁
+

(𝑥0 − 𝑥̅)2

∑(𝑥𝑖 − 𝑥̅)2
] 

Bu taxminiy dispersiyaning kvadrat ildizi prognozning standart xatosidir 

𝑠𝑒(𝑓) = √𝑣𝑎𝑟(𝑓)̂  

Quyidagi buyruq yordamida hisoblangan miqdor topiladi. 

predict sef, stdf 

Ma'lumotlarni ko'rib chiqamiz: 

 
yhat tuzilgan modelning hisoblangan qiymati va sef buyrug`i orqali prognozning 

standart xatosi qiymatlari hisoblanadi. Food_exp bog'liq o'zgaruvchi va eng kichik kvadrat 

qoldiqlari 41 kuzatuv uchun qiymatlar topilmagan. 

 



4.1.3. 𝒀̂ hisoblangan qiymat oralig'ini yaratish 

𝑡𝑐 kritik qiymatini t taqsimotidan 100(1 −
𝛼

2
)-protsentil deb belgilab, 100(1 − 𝛼)% 

prognozlash oralig'ini quyidagicha olishimiz mumkin. 

𝑦̂0 ± 𝑡𝑐𝑠𝑒(𝑓) 
Biz t-kritik qiymatni invttail(n,p) yordamida yaratamiz. 

scalar tc = invttail(38,.025) 

di "t critical value 97.5 percentile = " tc 

 
Hisoblangan qiymat - 2.0243942ga teng. So`ngra biz bu qiymatni yhat va sef bilan 

birgalikda bashorat qilish oralig'ining pastki va yuqori chegarasi bo'lgan yangi 

o'zgaruvchilarni yaratish uchun ishlatamiz. Yaratish buyrug'i gen ga qisqartirilib topish 

mumkin. 

gen lb = yhat - tc*sef 

gen ub = yhat + tc*sef 

Biz bu o'zgaruvchilar qiymatlarini 41-kuzatuvda ro'yxat yordamida ko'rishimiz ham 

mumkin. Biz list buyrug`i orqali 41-kuzatuvda income, lb, yhat va ub o'zgaruvchilarini 

miqdorini ifodalaymiz. 

list income lb yhat ub in 41 

 
Biz 41-kuzatuvni endi ishlatmaymiz, shuning uchun uni ma'lumotlar faylidan olib 

tashlaymiz. 

drop in 41 

drop buyrug`i bir nechta funktsiyalarga ega. Ma'lumotlar faylidan, biz yuqorida 

aytganimizdek, ba'zi kuzatuvlarni o'chirish uchun ishlatilishimiz mumkin va undan 

o'zgaruvchilarni tushirish uchun foydalanish mumkin. Sintaksisni ko'rish uchun help drop 

buyrug`iga to`xtalib qo`shimcha ma`lumotlarni olishingiz mumkin. 

 

4.2. Moslikni o'lchash 

Muvofiqlik o'lchovi 𝑅2 kvadratlarning parchalanish yig'indisi bilan izohlanadi. 

∑(𝑦𝑖 − 𝑦̅)2 =∑(𝑦̂𝑖 − 𝑦̅)2 +∑𝑒̂𝑖
2 

Bu "kvadratlar yig'indisi": 

-  ∑(𝑦𝑖 − 𝑦̅)2= kvadratlarning umumiy yig'indisi = SST: o'rtacha qiymat bo'yicha y 

ning umumiy o'zgarish o'lchovi. 

-  ∑(𝑦̂𝑖 − 𝑦̅)2= hisoblangan regressiya orqali kvadratlar yig'indisi = SSR: hisoblangan 

y qiymatining  umumiy y ning o`rtacha qiymatidan ayirmasining kvadratining summasini 

izohlanadi. Shuningdek, "hisoblangan kvadratlar yig'indisi" deb ham ataladi. 



- ∑ 𝑒̂𝑖
2 = xatolikning kvadratlar yig'indisi = SSE: regressiyada  izohlanmagan, y ning 

umumiy o'rtacha o'zgaruvchanligi. Kvadratlarning tushuntirilmagan yig'indisi, kvadrat 

qoldiqlar yig'indisi yoki kvadrat xatolar yig'indisi sifatida ham tanilgan. 

Ushbu qisqartmalar yordamida kvadratlarning parchalanishi yig'indisi hosil bo'ladi. 

𝑆𝑆𝑇 = 𝑆𝑆𝑅 + 𝑆𝑆𝐸 

U holda moslik o'lchovi 𝑅2 bo'ladi. 

𝑅2 =
𝑆𝑆𝑅

𝑆𝑆𝑇
= 1 −

𝑆𝑆𝐸

𝑆𝑆𝑇
 

Regressiya baholanganda, bu miqdorlarning barchasi regressiya koeffitsientlarining 

yuqori qismida Analysis of Variance Table paydo bo'ladi. Quyidagi rasmdagi ma`lumotlar 

(Model, Residual, Total) kvadratlarning parchalanish yig'indisi bilan bir xil bo'linishni 

anglatadi. SS deb belgilangan ustun kvadratlar yig'indisiga ishora qiladi. df ustuni erkinlik 

darajalari uchun. "Erkinlik darajalari" modeli 1 ga teng, chunki modelda bitta tushuntirish 

o'zgaruvchisi mavjud. Erkinlikning qoldiq darajalari N-2 ni tashkil qiladi, bu kuzatishlar 

sonidan model parametrlari soni ayirmasidir. MS ustuni Mean Square hisoblanadi. Ushbu 

ustunda SS ustunining df ustuniga bo'lingan nisbati mavjud. Shunday qilib, Residual MS - 

bu taxminiy xato dispersiyamizdir. 

 
 

4.2.1. Korrelyatsiya va determinatsiya 𝑹𝟐 

Shuningdek, Principle of Econometrics adabiyotida qayd etilishicha, oddiy regressiya 

modelida 𝑅2- y va x o'zgaruvchilar orasidagi korrelyatsion bog`lanishning kvadrati. Bundan 

tashqari, umuman olganda, 𝑅2  - y va uning hisoblangan qiymati 𝑦̂ o'rtasidagi kvadratik 

korrelyatsiyasi ham deb ataladi. Stata ning menyusidan foydalanib quyidagi ketma-ketlik 

asosida topiladi: 

Statistics > Summaries, tables, and tests > Summary and descriptive 

statistics > Pairwise correlations 

Pairwise correlations muloqot oynasida ochiladigan ro'yxat yordamida 

o'zgaruvchilar nomlarini kiriting. 

Ok tugmasini bosing. So`ngra Stata buyrug'i quyidagicha namoyon bo`ladi: 

pwcorr food_exp income yhat 

Shuningdek, natijalar oynasida Pirson korrelatsiya koeffitsienti namoyon bo`ladi. 



 
So`ngra food_exp va income o'rtasidagi kvadratik bog'liqlikni aniqlang. 

𝑟𝑥𝑦
2 = 0,622 = 0,385 = 𝑅2 

 

4.3. Ma'lumotlarni kattalashtirish va o'zgartirish ta`sirlari 

Regressiya modelidagi ma'lumotlar hech qanday haqiqiy ma'noni yo'qotmasdan 

kattalashtirish mumkin. O'zgaruvchilarning miqyosi o'ta katta yoki juda kichik sonlarsiz 

oqilona o'lchov birliklariga olib kelishi kerak. O'lchovli va o'zgartirilgan o'zgaruvchilar 

generate buyrug'i yordamida osongina yaratiladi. 

Oziq - ovqat xarajatlari to'g'risidagi ma'lumotlar uchun o'zgaruvchan daromad 100 

AQSh dollari birliklari bilan o'lchanadi, shuning uchun daromad = 20 xonadonning oylik 

daromadi 2000 AQSh dollarini tashkil etadi. Daromad o'zgaruvchisini miqdorini dollarga 

o'zgartirish uchun buyruqlar oynasiga quyidagini kiriting. 

gen inc_dollar = income*100 

Ushbu yangi daromad o'zgaruvchisi bilan oziq - ovqat xarajatlari regressiyasini 

hisoblang. E'tibor bering, biz regress buyrug'ini reg buyrug`iga qisqartirdik va quyidagi 

natijalarga erishdik: 

 
Regressiya natijalarida daromad koeffitsientining o'zgarishi va uning standart xatosi, 

shu bilan ishonch oralig'i qayd etiladi. 

log close 

 

4.3.1. Liner-log funktsiyasining shakli 

Chiziqli regressiya modeli o'zgaruvchilarning oddiy o'zgarishi yordamida chiziqli 

bo'lmagan o'zgaruvchilar o'rtasidagi munosabatni ifodalashi mumkin. Oziq-ovqat xarajatlari 

misoli uchun chiziqli-log modelini quyidagicha ifodalash mumkin: 

𝐹𝑂𝑂𝐷_𝐸𝑋𝑃 = 𝛽1 + 𝛽2 ln(𝐼𝑁𝐶𝑂𝑀𝐸) + 𝑒 

Ushbu modelni baholash uchun biz ln(income) bo'lgan yangi o'zgaruvchini yaratamiz 

va keyin eng kichik kvadrat regressiyasini qo'llaymiz. 

log using chap04_linlog, replace text 

use food, clear 



gen lincome = ln(income) 

reg food_exp lincome 

 
4.3.2-rejada biz foydalanadigan loglinear model regressidan mos qiymatlar va 

qoldiqlarni olish uchun taxmin qilishdan keyingi buyruq predict dan foydalanamiz. 

predict lyhat 

predict lehat, resid 

Chiziqli logli model uchun qiyalik 𝜷𝟐/𝑰𝑵𝑪𝑶𝑴𝑬 . Buni hisoblash uchun Stata 

dasturidan qanday foydalanishimiz mumkin? Avval daromad uchun qiymatni tanlashimiz 

kerak. Namunaviy o'rtacha qiymat, boshqa aniq qiymat qiziqtirmasa, yaxshi tanlov bo'ladi. 

Daromadlarning umumiy statistikasini hisoblaymiz. 

summarize income 

Stata hisoblangan bir necha natijalarni saqlaydi. Buni quyidagi buyruqni kiritish 

orqali ko'rishingiz mumkin: 

return list 

 
Yuqoridagi natijalarni ko'ramiz, unda bir qancha skalarlar ifodalanilgan, shu 

jumladan o'rtacha qiymat ham. Uni skalyar sifatida saqlashda quyidagi buyruqni kiriting. 

scalar xbar = r(mean) 

Endi slop formulasini hisoblash uchun lincom buyruqdan foydalaning. 

lincom lincome/xbar 



 
O'rtacha daromad darajasida har haftalik uy xo'jaliklari daromadi 100 dollarga 

oshishi, oziq -ovqat xarajatlarini 6,74 dollarga ko'payishiga olib keladi. Hisoblangan 

chiziqli-log modelning parametri boshqa qiymatlarda ham hisoblanishi mumkin. 

lincom lincome/10 

lincom lincome/20 

lincom lincome/30 

 

 
4.3.2. Hisoblangan liner-log modelini grafigini chizish 

Oldingi bo'limda hisoblangan qiymatlar va qoldiqlardan foydalanib, biz twoway 

grafiklardan foydalangan holda ikkita chizmani topamiz, birinchisi nuqtali chizma, 

ikkinchisi esa chiziqli chizma. Raqamlar chiroyli tarzda ifodalandi, chunki ma'lumotlar 

INCOME bo'yicha tartiblangan. 

twoway (scatter food_exp income, sort) /// 

(line lyhat income, sort lwidth(medthick)), /// 

xtitle(Income) ytitle(Food Expenditure) ylabel(0(100)600) /// 

title(Linear Log Model) 

graph save linlog, replace 



 
Bu munosabatlar to'g'ri chiziqli emas. Ma'lumotlar markaziga yaqin joyda o'rnatilgan 

chiziq juda tik emas. Koeffisient 6,74, oddiy chiziqli regressiyadan kichikroq. Buni aniqroq 

ko'rish uchun keling, uchta chizmanini ustiga qo'yamiz. Natijani topish va bashorat qilingan 

qiymatlar va qoldiqlarni olish uchun quietly variantni ishlatib, chiziqli munosabatlarni 

baholaymiz. 

quietly reg food_exp income 

predict yhat 

predict ehat, resid 

Mos keladigan chiziqni boshqa yangi buyruq bilan qo'shing, lpattern(dash). 

twoway (scatter food_exp income, sort) /// 

(line lyhat income, sort lwidth(medthick)) /// 

(line yhat income, sort lpattern(dash) lwidth(medthick)), /// 

xtitle(Income) ytitle(Food Expenditure) ylabel(0(100)600) /// 

title(Linear Log Model) 

graph save linlog2, replace 

 
 

4.3.3. Grafiklarni tahrirlash 

Graph Editor yordamida grafikning aspektlarini interaktiv tarzda o'zgartirish 

mumkin. 



 
O'ng tomonda batafsilroq ro'yxatga ochiladigan elementlarni bildiruvchi "+" bilan 

grafik elementlarning ro'yxati mavjud. Elementlarni bosish ular bilan bog'langan ajratilgan 

maydonlarni ko'rsatadi. Chiziqli-log modeli uchun “Hisoblangan qiymatlar” izohini 

oʻzgartirish uchun label[2] buyrugidan  foydalaning. 

 
label [2] ni tanlash chiziqni rangi, oʻlchami va matni uchun yuqori tahrirlash 

imkoniyatlarini koʻrsatadi. Faqat kerakli matnni kiriting va Enter tugmasini bosing holos. 

Plotregion va plot3ni tanlang. Plot3 tanlanganda, chiziqli o'rnatilgan chiziq 

tanlanganligini ko'rsatadigan kesishishlarni ko'rsatadi. Chiziq shaklini to`g`ri chiziqdan 

boshqasiga o'zgartirish uchun Stata ning menyusidan foydalaning. 

 
O'zgartirishlar kiritish uchun Apply-ni tanlang. Keyin o'zgartirishlaringizni Save 

orqali saqlang va Stop orqali plot muharririni to'xtating. 

 
Bizning grafik natijada quyidagicha ifodalaniladi: 



 
Graph Editor da grafikingizning deyarli barcha jihatlarini boshqarish mumkin. Siz 

kiritgan tahrirlarni graph recorder yordamida yozib olish va boshqa grafiklarda qayta 

qo`llash mumkin. Ushbu xususiyat haqida ko'proq ma'lumot olish uchun help graph 

recorder ga kirib o`rganishingiz mumkin. 

 

4.4. Qoldiqlarni tahlil olish 

Tanlangan funksional shaklning adekvatligini aniqlashning asosiy elementlaridan biri 

bu model qoldiqlarini tahlil qilishdir. Yuqoridagi 4.1.2-rejada chiziqli model uchun eng 

kichik kvadratlar qoldiqlarini hisoblab chiqdik. Ular xotiradagi ma'lumotlar to'plamida 

saqlab qoyilgandi. Agar yo'q bo'lsa, oziq-ovqat xarajatlarining chiziqli modelini qayta 

hisoblang va biz ehat deb ataydigan qoldiqlarni hisoblang. Agar sizda kengaytirilgan 

natijalar bo'lsa, keyinroq foydalanish uchun kerak bo'ladigan o'zgaruvchilarni saqlashni 

unutmang.  

Birinchi qadam sifatida eng kichik kvadratlar qoldiqlarining gistogrammasini tuzing. 

Biz har doim natija normal taqsimotni eslatib, qo'ng'iroq shaklida bo'lishiga umid qilamiz. 

histogram ehat, percent title(Linear Model Residuals) 

graph save olsehat_hist, replace 

 
Ikkinchidan, biz qoldiqlarni batafsil umumlashtirishimiz mumkin 

summarize ehat, detail 



 
 

4.4.1. Jarque-Bera testi 

Qoldiqlarni normal taqsimlanganini batafsil tahlil qilish uchun Jarque-Bera testidan 

foydalaniladi, bunda uning tarkibiy qismlari bo'lgan Skewness va Kurtosis koeffitsientlarini 

o'z ichiga oladi. Y o'zgaruvchisi uchun namuna jarayonlarini quyidagicha aniqlang: 

𝜇̃2 =∑(𝑌𝑖 − 𝑌̅)2/𝑁 = 𝜎̃2 

𝜇̃3 =∑(𝑌𝑖 − 𝑌̅)3/𝑁 

𝜇̃4 =∑(𝑌𝑖 − 𝑌̅)4/𝑁 

E'tibor bering, ushbu hisob-kitoblarni biz N ga bo'lamiz. Markaziy momentlarning 

ushbu namunaviy baholaridan foydalanib, biz skewness koeffitsienti (S) va kurtoz 

koeffitsienti (K) ning taxminiy qiymatini olishimiz mumkin. 

𝑠𝑘𝑒𝑤𝑛𝑒𝑠𝑠̂ = 𝑆 =
𝜇̃3
𝜎̃3

 

𝑘𝑢𝑟𝑡𝑜𝑠𝑖𝑠̂ = 𝐾 =
𝜇̃4
𝜎̃4

 

Yuqorida ta'kidlab o'tilganidek, umumlashtirishdan foydalanilganda, bir qator 

elementlar saqlanadi. Quyidagi buyruqni kiriting: 

return list 

Ba`zi elementlar takrorlanadi: 

 
Qaytarilgan N, skewness va kurtoz qiymatlaridan foydalanib, Jarque-Bera 

statistikasini hisoblash uchun Stata (yoki kalkulyator) dan foydalanishimiz mumkin. 

𝐽𝐵 =
𝑁

6
(𝑆2 +

(𝐾 − 3)2

4
) 



Statistikani yaratish va uni ko'rsatish uchun buyruqlar: 

scalar jb = (r(N)/6)*( (r(skewness)^2) + ((r(kurtosis)-3)^2)/4 ) 

di "Jarque-Bera Statistic = " jb 

Natijalar oynasida quyidagi ko`rinishda natija chiqadi: 

 
 

 

 

4.4.2. 𝒙𝟐-kvadrat taqsimotining kritik qiymatlari 

Ushbu taqsimot uchun kritik qiymat 2 daraja erkinlik bilan 𝑥2-kvadrat taqsimotidan 

kelib chiqadi. Kritik qiymatlar inv tipidagi funksiyalar yordamida hisoblanadi. To'g'ri 

funktsiya turini topish uchun help scalar, so'ngra biz oldin bir necha marta qilganimizdek, 

belgilash tugmasini bosamiz. Yana bir tezkor alternativ – kirish usulidir: 

db scalar 

"db" "muloqot oynasi" degan ma'noni anglatadi. Stata buyrug'i: 

scalar chic = invchi2tail(2,.05) 

di "Chi-square(2) 95th percentile = " chic 

Natijalar oynasida quyidagi hisoblangan qiymat: 

 
Hisoblangan Jarque-Bera statistikasining qiymati = 0,06334 testning kritik qiymati = 

5,99 dan ancha past, shuning uchun biz regressiya xatolari normal taqsimlangan degan 

gipotezani rad eta olmaymiz. 
 

4.4.3. 𝒙𝟐-kvadrat taqsimotining p-qiymatlari 

𝑥2-kvadrat kritik qiymatini olish o'rniga, biz Jarque-Bera test statistik qiymati uchun 

qiymatni olish orqali testni amalga oshirishimiz mumkin. To'g'ri funktsiya turini topish 

uchun db skaler dan foydalanamiz. Stata buyrug'I esa: 

scalar pvalue = chi2tail(2,jb) 

di "Jarque-Bera p-value = " pvalue 

Olingan p-qiymati 

 
p-qiymati 0,9688 ga teng bu esa I-turdagi xatolik 𝛼 = 0,05  ehtimolidan katta 

bo'lganligi sababli, biz avval aytib o'tganimizdek, normallikni rad eta olmaymiz. 

log close 

4.5. Polinomial modellar 

Polinom modeliga misol sifatida biz G'arbiy Avstraliyaning ba'zi fermalarida vaqt 

o'tishi bilan bug'doy hosili orasidagi bog`liqlikni ko'rib chiqamiz. Ma'lumotlar 

wa_wheat.dta deb nomlangan fayldan foydalanamiz Stata xotirasini tozalang va ushbu 

faylni ishga tushiring. O'zgaruvchilar tavsifi va umumiy statistikasi natijasini oling. 



log using chap04_wheat, replace text 

use wa_wheat, clear 

describe 

summarize 

 
Xulosa statistik ma'lumotlar shuni ko'rsatadiki, 4 xil fermada t=48 kuzatuv mavjud. 

Shunday qilib, bizning terminologiyamiz Principle of Econometrics, 4e nashirdagi adabiyot 

bilan parallel bo'lishi uchun, Greenough femasidagi ishlab chiqarishni o`rganish uchun 

YIELD deb nomlangan yangi o'zgaruvchini yaratamiz. 

gen yield = greenough 

label variable yield "wheat yield greenough shire" 

YIELD va TIME o'rtasidagi munosabatni ko'rsatadigan nuqtali grafikni tuzamiz. 

twoway (scatter yield time, sort) , /// 

xtitle(Time) ylabel(0(.5)2.5) ytitle(Yield) /// 

title(Wheat Yield) 

graph save wawheat, replace 

 
 

4.5.1. Chiziqli munosabatni baholash va tekshirish 

Bug'doy hosildorligining o'sish sur'ati, ayniqsa, davr oxiriga kelib ortgan. YIELD va 

TIME o'rtasidagi chiziqli munosabat mos kelmaydi, chunki u o'zgaruvchan nuqtalarga mos 

kelmaydi. Ko'rishimiz uchun YIELD va TIME o'rtasidagi chiziqli regressiyani hisoblaymiz 

va hisoblangan qiymat va eng kichik kvadratlar qoldiqlarini hisoblab olamiz. 

reg yield time 

predict yhat 

predict ehat, residuals 

Baholash natijalari quyidagicha bo`ladi: 



 
Hisoblangan eng kichik kvadratlar chizig'i va ma'lumotlarga mos chiziq bilan bir 

grafikda ifodalaymiz. 

twoway (scatter yield time, sort) /// 

(line yhat time, sort lwidth(medthick)) , /// 

xtitle(Time) ytitle(Yield) ylabel(0(.5)2.5) /// 

title(Wheat Yield Fitted Linear Model) 

graph save wheat_fit, replace 

 
Izoh: Uzoq qoldiqli oʻzgaruvchilar grafik ostidagi matnni oʻqib boʻlmasligiga olib 

kelishi mumkin. Grafik muharriri yordamida grafikni tahrirlang va yangi belgini kiriting. 

Amaliyot orqali siz vaqt davrining boshida va oxirida ijobiy qoldiqlar klasterini va 

o'rtadagi salbiy qoldiqlar klasterini aniqlay olasiz. Buni osonroq ko'rish uchun biz 

qoldiqlarni vaqtga nisbatan chizishimiz mumkin. Oddiy grafikdan foydalanib olish mumkin. 

twoway (scatter ehat time, sort) , /// 

xtitle(Time) ytitle(Residuals) yline(0) /// 

title(Wheat Linear Model Residuals) 

graph save wheat_ehat, replace 

yline(0) varianti gorizontal mos yozuvlar chizig'ini nolga tenglashtiradi. 



 
Shu bilan bir qatorda, Stata ba'zi o'rnatilgan qoldiq diagnostika grafiklariga ega. Stata 

menyu satrida quyidagilarni tanlang: 

 
help regress postestimation yordamida biz dialoglar va mavjud grafiklarning 

tavsiflariga havolalarni topa olamiz. 

 
Shuningdek, nuqtali grafikni o'rniga shtrixli grafikni olish uchun recast (bar) 

buyrug`idan foydalanish mumkin, 

rvpplot time, recast(bar) yline(0) 



 
Ushbu grafik vaqt davrining ikki oxiridagi ijobiy qoldiqlar to'plamini va markazdagi 

salbiy qoldiqlarning katta klasterini aniq ko'rsatadi. Qoldiqlardagi nuqtalar istalmagan va 

yaxshiroq funktsional shaklni topish zarurligini ko'rsatishi mumkin. 

 

4.5.2. Kubik tenglamani baholash va tekshirish 

time0=time/100 o'zgaruvchisini yarating. Bu o'lchov kub o'zgaruvchining juda katta 

bo'lishiga yo'l qo'ymaydi va taxminiy koeffitsientni nisbatan kattaroq qiladi. Kub 

tenglamani faktor oʻzgaruvchilari yozuvidan foydalanib baholang va oʻrnatilgan qiymatlar 

va qoldiqlarni oling, ularni yhat3 va ehat3 deb nomlang, chunki yhat va ehat oldingi 

regressiyada mavjud edi. 

generate time0=time/100 

list yield time0 in 1/5 

 
summarize time0 

 
reg yield c.time0#c.time0#c.time0 



 
Eng kichik kvadratlar qoldiqlari va taxmin qilingan qiymatlarni hisoblaymiz. 

predict yhat3 

predict ehat3, residuals 

Faktor belgilarining afzalligi shundaki, chekkalar biz uchun qiyalikni to'g'ri hisoblab 

chiqadi. 

margins, dydx(*) at(time=(0.15 0.30 0.45)) 

 
Ma'lumotlarning grafigi va o'rnatilgan chiziq kubik funktsiyaning egriligi davr 

boshida va oxirida ma'lumotlarga yaxshiroq mos kelishini ko'rsatadi. 

twoway (scatter yield time, sort) /// 

(line yhat3 time, sort lwidth(medthick)) , /// 

xtitle(Time) ytitle(Yield) ylabel(0(.5)2.5) /// 

title(Wheat Yield Fitted Cubic Model) 

graph save wheat_cubic_fit, replace 

 



Qoldiq diagnostika grafikdan foydalanib, biz ijobiy va salbiy qoldiqlarning 

qisqarishini aniqlaymiz. 

twoway (scatter ehat3 time, sort) , /// 

xtitle(Time) ytitle(Residuals) yline(0) /// 

title("Residuals Wheat" "Cubic Specification") 

graph save wheat_cube_ehat, replace 

 
 

4.5.3. Hosildorlikning log-liner o'sish modelini baholash 

Oldingi bo'limning kubik modeliga muqobil ravishda log-chiziqli o'sish modelidan 

foydalanish ham mumkin. 

gen lyield = ln(yield) 

reg lyield time 

 
Biz yillik o'sish sur'atini 1,78% deb hisoblaymiz. log-chiziqli funktsional shakl haqida 

keyingi bo'limda yanada chuqurroq to`xtalib o`tamiz. 

log close 

 

4.6. Log-liner oylik maosh tenglamasini hisoblash 

Yangi faylda ishlashni boshlang, bunda cps4_small.dta ma'lumotlar faylidan 

foydalaning. Shuningdek, o'zgaruvchilar tavsifini ifodalang  va tasviriy statistikani toping. 

EDUC qiymatlarini jadvalda ifodalang. 

log using chap04_lwage, replace text 

use cps4_small, clear 



describe 

summarize 

tabulate educ 

Biz foydalanadigan o'zgaruvchilar quyidagilar: 

 
msize(small)  varianti bilan kichik belgilardan foydalangan holda ish haqining ta'lim 

yillariga nisbatan nuqtali grafigini tuzamiz. 

twoway (scatter wage educ, msize(small)) , /// 

xtitle(Education) ytitle(Wage) /// 

title(Wage-Education Scatter) 

graph save wage_educ, replace 

 
Ushbu ma'lumotlardan foydalanib, biz ish haqi va ta'lim yillari o'rtasidagi bog'liqlikni 

o'rganamiz. Ish haqining ta'limga qarshi grafigi ta'limning yuqori darajalarida ba'zi ish 

haqining tarqoqlik tendentsiyasini ko'rsatadi. Bu turdagi maoshlar, ish haqi va daromad 

ma'lumotlariga xosdir. 

 

4.6.1. Log-liner modeli 

Bu log-chiziqli model, ya'ni ta`sir etuvchi o'zgaruvchini o'zgartirish orqali amalga 

oshiriladi. 

gen lwage = ln(wage) 

Endi ta'limga qarshi ln(wage) nuqtali grafigni tayyorlaymiz. O'zgartirilgan 

ma'lumotlar sezilarli darajada kamroq tarqoqlikga ega. 

twoway (scatter lwage educ, msize(small)), /// 

xtitle(Education) ytitle(ln(Wage)) /// 

title(ln(Wage)-Education Scatter) 

graph save lwage_educ, replace 



 
Ushbu misolda biz 12 yillik ta`limdagi ma'lumotga ega bo'lgan shaxslarning ish 

haqini taxminiy hisoblaymiz. Shunday qilib, 1001 kuzatuviga EDUC = 12 qiymatini 

qo'shish orqali ma'lumotlar faylini tahrirlaymiz. 

edit 

set obs 1001 

replace educ=12 in 1001 

Log-chiziqli regressiyani hisoblaymiz va husoblangan qiymatlarni, eng kichik 

kvadratlar qoldiqlarini va prognozning standart xatosini hisoblaymiz. 

reg lwage educ 

 
Hisoblangan qiymatlarni, qoldiqlarni va kelajakda foydalanish uchun prognozning 

standart xatosini hisoblaymiz. 

predict lwagehat 

predict ehat, residuals 

predict sef, stdf 

Regressiya natijalariga ko'ra, biz har bir qo'shimcha ta'lim yili ish haqining taxminan 

9% oshishiga olib kelishini taxmin qilishimiz mumkin. 

Quyida biz hisoblashda xato atamasining taxminiy dispersiyasidan foydalanamiz, 

shuning uchun regressiya natijalari Stata xotirasida bo'lsa, biz undan foydalanamiz. 

Regressiyadan keyin Stata tomonidan saqlangan elementlar ro'yxatini ko'rish uchun ereturn 

list buyrug`idan foydalanamiz. 

ereturn list 



 
Qaytarilgan skalarlar orasida erkinlik darajalari (df_r) va kvadrat qoldiqlar yig'indisi 

(ssr) mavjud. Xato atamasining taxminiy dispersiyasi bo'lgan skalarni hisoblaymiz. 

scalar sig2 = e(rss)/e(df_r) 

di "sigma-hat squared = " sig2 

 
Endi oldingi kabi buyruqlardan foydalanib, gistogramma tuzamiz va Jarque-Bera 

testini normallik uchun tekshiramiz.  

histogram ehat, percent title(log(Wage) Model Residuals) 

graph save lwage_ehat, replace 

 
summarize ehat, detail 

scalar jb = (r(N)/6)*( (r(skewness)^2) + ((r(kurtosis)-3)^2)/4 ) 

di "Jarque-Bera Statistic = " jb 

scalar chic = invchi2tail(2,.05) 

di "Chi-square(2) 95th percentile = " chic 

scalar pvalue = chi2tail(2,jb) 

di "Jarque-Bera p-value = " pvalue 



 
Natijada qoldiqlarning normalligi rad etiladi. So`ngra qoldiqlarning taqsimlanish 

yordamida grafikni olamiz. 

rvpplot educ, yline(0) 

 
Qoldiqlar ham pasayish tendentsiyasini ifodalayapti, bu esa ta'lim oshgani sayin ish 

haqi oshishini ko'rsatadi. 

 

4.6.2. Hisoblangan ish haqlarini hisoblash 

Log-chiziqli modeldan hisoblangan qiymat ln(𝑦) = 𝑏1 + 𝑏2𝑥̂  dir. 𝑦  bog'liq 

o'zgaruvchining bashoratini olish uchun biz "natural logorifimlangan" qiymatini olgan holda 

logarifimlangan qiymatini olamiz. 

𝑦̂𝑛 = exp⁡(ln⁡(𝑦))̂ = exp⁡(𝑏1 + 𝑏2𝑥) 
Stata ning exp() funktsiyasi eksponensial funktsiyani ifodalaydi. Hisoblangan 

ln(wage) dan foydalanish uchun logorifimlangan o`zgaruvchisini  yaratamiz. 

gen yhatn = exp(lwagehat) 

Ko`pchilik namunalarda yaxshiroq hisoblangan qiymatga erishish uchun 

"to`g`rilangan" deb nomlanuvchi hisoblangan qiymatlar hisoblanadi. 

𝑦̂𝑐 = 𝐸(𝑦)̂ = exp(𝑏1 + 𝑏2𝑥 +
𝜎̂2

2
) = 𝑦̂𝑛𝑒

𝜎̂2/2 

Eslatib o'tamiz, biz hisoblangan xato dispersiyasi 𝜎̂2 ni sig2 deb hisoblab chiqdik. 

Ushbu koeffisientni “tuzatish koeffitsienti” hisoblangan qiymatlar yordamida hosil qilish 

mumkin. 

di "correction factor = " exp(sig2/2) 



 
gen yhatc = yhatn*exp(sig2/2) 

Tuzatish koeffitsienti 1,1487 ni tashkil qildi va shuning uchun "tuzatilgan" 

hisoblangan qiymatlar "tabiiy" qiymatlardan kattaroqdir. Bu har doim shunday bo'ladi, 

chunki har qanday 𝑎 > 0  qiymati uchun tuzatish koeffitsienti 𝑒𝑎 > 1  va taxminiy 

dispersiya 𝜎̂2  har doim noldan katta. 

 

4.6.3. Ish haqi grafiklarini tuzish 

Ikki hisoblangan qiymatlarni taqqoslaydigan grafikda ko'rish yanada yaxshi xulosalar 

chiqarish mumkin. 

twoway (scatter wage educ, sort msize(small)) /// 

(line yhatn educ, sort /// 

lwidth(medthick) lpattern(dash)) /// 

(line yhatc educ, sort lwidth(medthick) lpattern(solid)) 

graph save lwage_predict, replace 

 
Grafik yhatc har doim yhatn dan katta ekanligini ko'rsatadi. Tabiiy va hisoblangan 

qiymatlarni ko'rish uchun biz 1001 kuzatuvini qiymatini ko`rib o`tamiz. 

list educ yhatn yhatc in 1001 

 
Bu qiymatlar mantiqiymi? Bizning namunamizda 12 yillik ta`lim ma'lumotga ega 

bo'lganlar uchun ish haqi bo'yicha umumiy statistikani hisoblash orqali biz taxminiy 

tekshirishimiz mumkin. 

summarize wage if educ==12 in 1/1000 

 



Biz prognozlar 12 yillik ta'limga ega bo'lgan shaxslarning haqiqiy o'rtacha ish haqiga 

mos kelishini ko'rishimoz mumkin. 

 

4.6.4. Determinatsiya koeffisienti 𝑹𝟐 

Determinatsiya koeffisienti 𝑅2  o'lchovi "eng yaxshi" hisoblangan qiymatni 

baholovchi yhatc va o'zgaruvchan income o'rtasidagi korrelyatsiya kvadratidir. Bu holda 

yhatc va yhatn faqat multiplikativ konstanta bilan farqlanadi, shuning uchun ularning 

income bilan bog'liqligi bir xil bo'ladi. Korrelyatsiyalarni hisoblash uchun quyidagi 

buyruqdan foydalaning: 

correlate wage yhatn yhatc 

di "r2g = " r(rho)^2 

 
 

 

4.6.5. Log-liner modeldagi bashorat oraliqlari 

Log-chiziqli modeldagi bashorat oralig'i sifatida quyidagi formula tuzilgan. 

[exp (ln(𝑦)̂ − 𝑡𝑐𝑠𝑒(𝑓)) , exp (𝑙𝑛(𝑦)̂ + 𝑡𝑐𝑠𝑒(𝑓))] 

U "tabiiy" hisoblangan qiymatga va yuqoridagi 4.6.1-bo'limda tuzilgan prognozning 

standart xatosiga asoslanadi. Birinchidan, 998 daraja erkinlik bilan t-taqsimotidan 97,5 foizli 

ehtimollikni hisoblaymiz. 

scalar tc = invttail(998,.025) 

ln(income) ning pastki va yuqori prognoz oraliqlarini topamiz. 

gen lb_lwage = lwagehat - tc*sef 

gen ub_lwage = lwagehat + tc*sef 

Eksponensial funktsiyadan foydalanib logorifim qiymatlarini topamiz. 

gen lb_wage = exp(lb_lwage) 

gen ub_wage = exp(ub_lwage) 

1001-kuzatish uchun bashorat oralig'ining qiymatlarini ko`rib o'tamiz. 

list lb_wage ub_wage in 1001 

 
Ish haqi ma'lumotlari, tabiiy hisoblangan qiymat va ta'limga qarshi intervalli 

hisoblangan qiymat bilan grafik yaratamiz. 



twoway (scatter wage educ, sort msize(small)) /// 

(line yhatn educ, sort lwidth(medthick) lpattern(solid)) /// 

(line ub_wage educ, sort lcolor(forest_green) lwidth(medthick) /// 

lpattern(dash)) /// 

(line lb_wage educ, sort lcolor(forest_green) lwidth(medthick) /// 

lpattern(dash)) 

graph save lwage_interval, replace 

 
Natijalarga erishilgandan so`ng ishchi fayilni yopish tavxiya qilinadi va buni quyidagi 

buyruq orqali amalga oshiring. 

log close 

 

4.7. Log-log modeli 

Log-log funksiyasi ln(𝑦) = 𝛽1 + 𝛽2ln⁡(𝑥)  talab tenglamalari va ishlab chiqarish 

funksiyalarini hisoblash uchun keng qo‘llaniladi. "Log-log" nomi logarifm tenglamaning 

har ikki tomonida paydo bo'lishidan kelib chiqadi. Ushbu modeldan foydalanish uchun 𝑦⁡va 

𝑥 ning barcha qiymatlari ijobiy bo'lishi kerak. Ushbu egri chiziqlarning qiyaliklari har bir 

nuqtada o'zgaradi, lekin elastiklik doimiy va 𝛽2  ga teng. Log-log funksiyasi haqida fikr 

yuritishning foydali usuli uning elastikligi 
𝑑𝑦

𝑑𝑥
= 𝛽2 (

𝑦

𝑥
)  ni yaqinroq tekshirishdan kelib 

chiqadi.  

Buni 𝛽2 = (
𝑑𝑦

𝑦
)/(

𝑑𝑥

𝑥
)⁡  ga o'zgartiring. Shunday qilib, log-log funktsiyasining 

elastikligi doimiy nisbiy o'zgarishlarni ko'rsatadi, chiziqli funktsiya esa doimiy mutlaq 

o'zgarishlarni ko'rsatadi. Yangi ishchi faylini oching va newbroiler.dta ma'lumotlaridan 

foydalaning. 

log using chap04_loglog, replace text 

use newbroiler, clear 

describe 

summarize 

Bunda sifat o'zgaruvchilarni ham o'z ichiga olgan. 



 
Miqdor va narxning logarifmini tuzing va log-log modelini baholang. Bu yerda logga 

ekvivalent bo'lgan ln log funktsiyasidan foydalanamiz. 

gen lq = ln(q) 

gen lp = ln(p) 

reg lq lp 

 
Hisoblangan elastiklik -1,12 ga teng hisoblanadi. 

Hisoblangan qiymatlar va grafikni tuzamiz. Plot buyruqlari biroz uzun, shuning uchun 

biz ularni bu yerda ikkinchi qatorga ifodaladik, lekin uni do-faylda ifodalamaymiz. 

predict lqhat 

scalar sig2 = e(rss)/e(df_r) 

gen qhatc = exp(lqhat)*exp(sig2/2) 

twoway (scatter q p, sort msize(small) lwidth(medthick) /// 

lpattern(solid)) /// 

(line qhatc p, sort lwidth(medthick)), /// 

xtitle(Price of Chicken) ytitle(Quantity of Chicken) /// 

title(Poultry Demand) 



 
Determinatsiya koeffisienti 𝑅2 ni hisoblaymiz. 

correlate q qhatc 

di "r2g = " r(rho)^2 

log close 

 
IV bob mavzularini mustahkamlash uchun savollar 

1. Regressiyani baholash uchun qanday amallar ketma ketligi bajarilishi kerak? 

2. 𝑌̂ hisoblangan qiymat oralig'ini yaratish buyrug’i qanday? 

3. drop buyrug`i funksiyalarini sanang. 

4. Muvofiqlik o'lchovini ifodalang. 

5. oddiy regressiya modelida 𝑅2 nimani ifodalaydi? 

6. O'lchovli va o'zgartirilgan o'zgaruvchilar qanday buyruq orqali yaratiladi? 

7. Hisoblangan liner-log modelini grafigini chizish ketma ketligini yozing. 

8. Grafik yaratilganidan so’ng uni qayta tahrirlash qanday amalga oshiriladi? 

9. Jarque-Bera testi vazifasi nima? 

10.  Kub tenglamani faktor oʻzgaruvchilari yozuvidan foydalanib baholash qanday ketma ketlik asosida 

amalga oshiriladi? 

11.  Hisoblangan Ish haqi grafiklarini tuzishni amalda bajaring. 

12.  Log-log funksiyasi qay hollarda qo’llaniladi? 

 


